Contractions!

```
c_{x_,y_}[w_Wedge] := Module[{i, j},
  {i} = FirstPosition[w, x, {0}]; {j} = FirstPosition[w, y, {0}];
  { w                                              (i == 0) ∧ (j == 0)
  { (-1)^(i+j+If[i>j,0,1]) Delete[w, {{i}, {j}}]   (i > 0) ∧ (j > 0)
];
c_{x_,y_}[ℰ_] := ℰ /. w_Wedge ⧴ c_{x,y}[w]
```

```
WExp[a ∧ b + 2 c ∧ d]
c_{d,c}@WExp[a ∧ b + 2 c ∧ d]
```

```
Wedge[] + a ∧ b + 2 c ∧ d + 2 a ∧ b ∧ c ∧ d
-Wedge[] - a ∧ b
```

$\mathcal{A}$`[is,os,cs,w]` is also a container for the values of the $\mathcal{A}$-invariant of a tangle. In it, `is` are the labels of the input strands, `os` are the labels of the output strands, `cs` is an assignment of colours (namely, variables) to all the ends $\{\xi_i\}_{i\in \mathtt{is}} \sqcup \{x_j\}_{j\in \mathtt{os}}$, and `w` is the "payload": an element of $\Lambda(\{\xi_i\}_{i\in \mathtt{is}} \sqcup \{x_j\}_{j\in \mathtt{os}})$.



$X_{ijkl}[S,T]$

```
𝒜[X_{i_,j_,k_,l_}[S_, T_]] := 𝒜[{l, i}, {j, k}, ⟨|ξ_i → S, x_j → T, x_k → S, ξ_l → T|⟩,
   Expand[T^{-1/2} WExp[Expand[{ξ_l, ξ_i}.(1 1-T; 0 T).{x_j, x_k}] /. ξ_a_ x_b_ ⧴ ξ_a ∧ x_b]]];
𝒜[X_{1,2,3,4}[u, v]]
```

```
𝒜[{4, 1}, {2, 3}, ⟨|ξ1 → u, x2 → v, x3 → u, ξ4 → v|⟩,
  Wedge[]/√v - (x2 ∧ ξ4)/√v - √v x3 ∧ ξ1 - (x3 ∧ ξ4)/√v + √v x3 ∧ ξ4 + √v x2 ∧ x3 ∧ ξ1 ∧ ξ4]
```

```
𝒜[X_{i_,j_,k_,l_}] := 𝒜[X_{i,j,k,l}[τ_i, τ_l]]
```

The negative crossing and the "point":



$\bar{X}_{ijkl}[S,T]$ $P_{ij}[T]$

```
𝒜[X̄_{i_,j_,k_,l_}[S_, T_]] := 𝒜[{i, j}, {k, l}, ⟨|ξ_i → S, ξ_j → T, x_k → S, x_l → T|⟩,
   Expand[T^{1/2} WExp[Expand[{ξ_i, ξ_j}.(T^-1 0; 1-T^-1 1).{x_k, x_l}] /. ξ_a_ x_b_ ⧴ ξ_a ∧ x_b]]];
𝒜[X̄_{i_,j_,k_,l_}] := 𝒜[X̄_{i,j,k,l}[τ_i, τ_j]];
𝒜[P_{i_,j_}[T_]] := 𝒜[{i}, {j}, ⟨|ξ_i → T, x_j → T|⟩, WExp[ξ_i ∧ x_j]];
𝒜[P_{i_,j_}] := 𝒜[P_{i,j}[τ_i]]
```

The linear structure on $\mathcal{A}$'s:

```
𝒜 /: α_ × 𝒜[is_, os_, cs_, w_] := 𝒜[is, os, cs, Expand[α w]]
𝒜 /: 𝒜[is1_, os1_, cs1_, w1_] + 𝒜[is2_, os2_, cs2_, w2_] /;
   (Sort@is1 == Sort@is2) ∧ (Sort@os1 == Sort@os2) ∧
    (Sort@Normal@cs1 == Sort@Normal@cs2) := 𝒜[is1, os1, cs1, w1 + w2]
```

Deciding if two $\mathcal{A}$'s are equal:

```
𝒜 /: 𝒜[is1_, os1_, _, w1_] ≡ 𝒜[is2_, os2_, _, w2_] :=
  TrueQ[(Sort@is1 === Sort@is2) ∧ (Sort@os1 === Sort@os2) ∧
    PowerExpand[w1 == w2]]
```

The union operation on $\mathcal{A}$'s (implemented as "multiplication"):




```
𝒜 /: 𝒜[is1_, os1_, cs1_, w1_] × 𝒜[is2_, os2_, cs2_, w2_] :=
  𝒜[is1 ⋃ is2, os1 ⋃ os2, Join[cs1, cs2], WP[w1, w2]]
Short[𝒜[X_{2,4,3,1}[S, T]] × 𝒜[X̄_{3,4,6,5}], 5]
```

```
𝒜[{1, 2, 3, 4}, {3, 4, 5, 6},
 ⟨|ξ2 → S, x4 → T, x3 → S, ξ1 → T, ξ3 → τ3, ξ4 → τ4, x6 → τ3, x5 → τ4|⟩, (√τ4 Wedge[])/√T -
  (√τ4 x3 ∧ ξ1)/√T + √T √τ4 x3 ∧ ξ1 - √T √τ4 x3 ∧ ξ2 - (√τ4 x4 ∧ ξ1)/√T - (√τ4 x5 ∧ ξ4)/√T -
  (x6 ∧ ξ3)/(√T √τ4) + <<40>> + (√T x3 ∧ x5 ∧ x6 ∧ ξ1 ∧ ξ3 ∧ ξ4)/√τ4 - (√T x3 ∧ x5 ∧ x6 ∧ ξ2 ∧ ξ3 ∧ ξ4)/√τ4 -
  (x4 ∧ x5 ∧ x6 ∧ ξ1 ∧ ξ3 ∧ ξ4)/(√T √τ4) + (√T x3 ∧ x4 ∧ x5 ∧ x6 ∧ ξ1 ∧ ξ2 ∧ ξ3 ∧ ξ4)/√τ4]
```

Contractions of $\mathcal{A}$-objects:

```
c_{h_,t_}@𝒜[is_, os_, cs_, w_] := 𝒜[
   DeleteCases[is, t], DeleteCases[os, h], KeyDrop[cs, {x_h, ξ_t}], c_{x_h,ξ_t}[w]
  ] /. If[MatchQ[cs[ξ_t], τ_], cs[ξ_t] → cs[x_h], cs[x_h] → cs[ξ_t]];
c_{4,4}[𝒜[X_{2,4,3,1}[S, T]] × 𝒜[X̄_{3,4,6,5}]]
```

```
𝒜[{1, 2, 3}, {3, 5, 6}, ⟨|ξ2 → S, x3 → S, ξ1 → T, ξ3 → τ3, x6 → τ3, x5 → T|⟩,
  Wedge[] - x3 ∧ ξ1 + T x3 ∧ ξ1 - T x3 ∧ ξ2 - x5 ∧ ξ1 - x6 ∧ ξ1 + (x6 ∧ ξ1)/T - (x6 ∧ ξ3)/T +
   T x3 ∧ x5 ∧ ξ1 ∧ ξ2 - x3 ∧ x6 ∧ ξ1 ∧ ξ2 + T x3 ∧ x6 ∧ ξ1 ∧ ξ2 + x3 ∧ x6 ∧ ξ1 ∧ ξ3 -
   (x3 ∧ x6 ∧ ξ1 ∧ ξ3)/T - x3 ∧ x6 ∧ ξ2 ∧ ξ3 - (x5 ∧ x6 ∧ ξ1 ∧ ξ3)/T - x3 ∧ x5 ∧ x6 ∧ ξ1 ∧ ξ2 ∧ ξ3]
```

Automatic and intelligent multiple contractions:

```
c@𝒜[is_, os_, cs_, w_] := Fold[c_{#2,#2}[#1] &, 𝒜[is, os, cs, w], is ⋂ os]
𝒜[{A_𝒜}] := c[A];
𝒜[{A1_𝒜, As__𝒜}] := Module[{A2},
  A2 = First@MaximalBy[{As}, Length[A1⟦1⟧ ⋂ #⟦2⟧] + Length[A1⟦2⟧ ⋂ #⟦1⟧] &];
  𝒜[Join[{c[A1 A2]}, DeleteCases[{As}, A2]]] ]
𝒜[𝒪s_List] := 𝒜[𝒜 /@ 𝒪s]
c[𝒜[X_{2,4,3,1}[S, T]] × 𝒜[X̄_{3,4,6,5}]]
```

```
𝒜[{1, 2}, {5, 6}, ⟨|ξ2 → S, ξ1 → T, x6 → S, x5 → T|⟩,
 Wedge[] - x5 ∧ ξ1 - x6 ∧ ξ2 - x5 ∧ x6 ∧ ξ1 ∧ ξ2]
```

```
𝒜@{𝒜[X_{2,4,3,1}[S, T]], 𝒜[X̄_{3,4,6,5}]}
```

```
𝒜[{1, 2}, {5, 6}, ⟨|ξ2 → S, ξ1 → T, x6 → S, x5 → T|⟩,
 Wedge[] - x5 ∧ ξ1 - x6 ∧ ξ2 - x5 ∧ x6 ∧ ξ1 ∧ ξ2]
```

4. Skein relations and evaluations for $\mathcal{A}$




```
𝒜@{X̄_{4,1,6,3}[v, u], X̄_{3,2,5,4}}
```

```
𝒜[{1, 2}, {5, 6}, ⟨|ξ2 → v, x5 → u, ξ1 → u, x6 → v|⟩,
 √u √v Wedge[] - (√u x5 ∧ ξ1)/√v + (√u x5 ∧ ξ2)/√v - √u √v x5 ∧ ξ2 + (√v x6 ∧ ξ1)/√u - √u √v x6 ∧ ξ1
  (√v x6 ∧ ξ2)/√u - (√u x5 ∧ x6 ∧ ξ1 ∧ ξ2)/√v - (√v x5 ∧ x6 ∧ ξ1 ∧ ξ2)/√u + √u √v x5 ∧ x6 ∧ ξ1 ∧ ξ2]
```

Video and more at `http://www.math.toronto.edu/~drorbn/Talks/MoscowByWeb-2104/`